

- **Subsets in DATA step in a macro - use IF statements**

The usual SAS IF statements, including the subsetting IF, **will** work inside a SAS macro. This allows you to select subsets of data sets within a macro. Here is an example using the flowers dataset in homework 5, saved as a permanent SAS dataset in the Sasuser library.

```
%let type1 = Ginger;
%let type2 = Protea;
%macro rod;
  data temp1;
  set sasuser.flowers;
  if variety = "&type1";
  run;

  data temp2;
  set sasuser.flowers;
  if variety = "&type2";
  run;

  proc print data = temp1;
  title "flower data for &type1";
  run;

  proc print data = temp2;
  title "flower data for &type2";
  run;

%mend rod;

%rod
```

Note: You can use a regular subsetting "if" statement within a macro.

- **Subsets in PROC PRINT within a macro - use WHERE**

```
%let type1 = Ginger;
%let type2 = Protea;

%macro rod;
  data temp;
  set sasuser.flowers;
  run;

  proc print data = temp;
  title "flower data for &type1";
  where variety = "&type1";
```

```
run;

proc print data = temp;
title "flower data for &type2";
where variety = "&type2";
run;

%mend rod;

%rod
```

Note: You can use a regular "where" statement in a PROC PRINT within a macro to print a subset of the data.

- **Why use %IF and %THEN when the usual if and then statements work within macros?**

(Delwiche and Slaughter p 208) You may be wondering why anyone needs these statements. Why not just use the standard IF-THEN? You may indeed use standard IF-THEN statements in your macros, but you will use them for different actions. %IF statements can contain actions that standard IF statements can't contain, such as complete DATA or PROC steps and even other macro statements. The %IF-%THEN statements don't appear in the standard SAS code generated by your macro.

- **Example of using %IF %THEN with different data and proc steps**

There are two SAS datasets, sasuser.tumor and sasuser.corr1 stored in the Sasuser library.

```
%macro rod(choose);

%IF &choose = tumor %THEN %DO;
data temp;
set sasuser.tumor;
run;

proc print data = sasuser.tumor;
title "Choosing the tumor dataset";
run;

%END;

%ELSE %IF &choose = corr1 %THEN %DO;

data temp;
set sasuser.corr1;
```

```
run;

proc corr data = sasuser.corr1;
  title "Choosing the corr1 dataset";
  var x y;
  run;
%END;

%MEND rod;

%rod(tumor)
%rod(corr1)
```